TEMPLATE = lib
QT      -= gui
TARGET   = cpcap-1.0
CONFIG  += dll

SOURCES = \ 
    devices.cpp \
    devices-info.cpp \
    capture-device.cpp \
    helpers.cpp

unix:SOURCES += linux-fcn.cpp
win32:SOURCES += sample-grabber.cpp

HEADERS = \
    include/cpcap-device-enumerator.hpp \
    include/common.hpp \ 
    include/devices.hpp \
    include/cpcap-capture-devices.hpp \
    include/platform.hpp

win32:HEADERS += include/mdshow.hpp \
    include/sample-grabber.hpp
unix:HEADERS += include/linux-fcn.hpp

INCLUDEPATH = include

DEFINES += USE_QOBJECT
unix:DEFINES  += LINUX
win32:DEFINES += WIN32 BUILD_CPCAP_DLL _CRT_SECURE_NO_WARNINGS _CRT_SECURE_NO_DEPRECATES

CONFIG(debug, release|debug) {
    DEFINES += DEBUG
}

libheaders.files        = include/CPCap $$HEADERS
unix:libheaders.path    = /usr/include/libcpcap-1.0
unix:target.path        = /usr/lib

win32:libheaders.path  = C:/libcpcap-1.0/include/libcpcap-1.0
win32:target.path      = C:/libcpcap-1.0/bin

INSTALLS = libheaders target
win32:LIBS = -lstrmiids -lole32 -loleaut32
linux-g++:target.path = /usr/lib
win32:target.path = C:/trazoscopio-install
INSTALLS = target

MOC_DIR = moc
OBJECTS_DIR = obj
DESTDIR = construido
